



**HAL**  
open science

# Towards A Certified Version of the Encyclopedia of Triangle Centers

Julien Narboux, David Braun

► **To cite this version:**

Julien Narboux, David Braun. Towards A Certified Version of the Encyclopedia of Triangle Centers. Mathematics in Computer Science, 2016, 10 (1), pp.17. 10.1007/s11786-016-0254-4 . hal-01174131v2

**HAL Id: hal-01174131**

**<https://inria.hal.science/hal-01174131v2>**

Submitted on 24 Feb 2016

**HAL** is a multi-disciplinary open access archive for the deposit and dissemination of scientific research documents, whether they are published or not. The documents may come from teaching and research institutions in France or abroad, or from public or private research centers.

L'archive ouverte pluridisciplinaire **HAL**, est destinée au dépôt et à la diffusion de documents scientifiques de niveau recherche, publiés ou non, émanant des établissements d'enseignement et de recherche français ou étrangers, des laboratoires publics ou privés.

Copyright

# Towards A Certified Version of the Encyclopedia of Triangle Centers

Julien Narboux and David Braun

**Abstract.** Triangle centers such as the center of gravity, the circumcenter, the orthocenter are well studied by geometers. Recently, under the guidance of Clark Kimberling, an electronic encyclopedia of triangle centers (ETC) has been developed, it contains more than 7000 centers and many properties of these points. In this paper, we describe how we created a certified version of ETC such that some of the properties described come along with a computer checked proof of its validity.

**Mathematics Subject Classification (2010).** Primary 51-04; Secondary 97G99.

**Keywords.** triangle centers, geometry, formal proof, proof assistant, Coq .

## Introduction

The characteristic points of triangles such as the center of gravity, the circumcenter, the orthocenter have been well studied for centuries by geometers. Some properties of these points are well known. For example the three points we cited are collinear, they belong to a line called the Euler line. More recently, under the guidance of Clark Kimberling [Kim93, Kim94, Kim98], an electronic encyclopedia of triangle center (ETC) has been developed<sup>1</sup>. This encyclopedia contains currently the definition of more than 7000 triangle centers as well as the description of many properties of these points. Similarly, an encyclopedia of more than 700 cubics is maintained by Bernard Gibert<sup>2</sup> and an encyclopedia of quadri-figures is maintained by Chris van Tienhoven<sup>3</sup>.

The ETC consists in web pages containing hundreds of thousands of properties. But most of the results are presented **without proof** nor reference. A large majority of the results have been generated by computer programs using symbolic and/or numeric manipulations. But the source code of the computations are not given in the encyclopedia so the reader can not reproduce the results and we can not verify if the result has been found by symbolic or numeric computations. How can one trust these results ? What are the *potential* sources of errors ?

1. There could be an error in the definition of the points in the Computer Algebra System (CAS)<sup>4</sup>.
2. There could be an error in the definition of the algorithms to check properties implemented in the language of the CAS.
3. There could be a bug in the implementation of the CAS itself or in the theory behind the algorithms used.

---

ICube, UMR 7357 CNRS, University of Strasbourg.

<sup>1</sup><http://faculty.evansville.edu/ck6/encyclopedia/ETC.html>

<sup>2</sup><http://bernard.gibert.pagesperso-orange.fr/>

<sup>3</sup><http://www.chrisvantienhoven.nl/index.php/mathematics/encyclopedia>

<sup>4</sup>For example the definitions of points  $X_{5598}$  to  $X_{5602}$  contained an error in the version of ETC as of May 2015, the point was defined as  $a^4 - a^2(b+c)^2 - 4aS\sqrt{r} + 4R$  instead of  $a^4 - a^2(b+c)^2 - 4aS\sqrt{rR} + 4R^2$

Number	Name	Trilinear Coordinates	Barycentric Coordinates
$X_1$	Incenter	1	$a$
$X_2$	Centroid	$1/a$	1
$X_3$	Circumcenter	$\cos(A)$	$a^2(a^2 - b^2 - c^2)$
$X_4$	Orthocenter	$\sec(A)$	$(a^2 + b^2 - c^2)(a^2 - b^2 + c^2)$
$X_5$	Nine-point center	$\cos(B - C)$	$a^2b^2 - b^4 + a^2c^2 + 2b^2c^2 - c^4$

TABLE 1. Some ETC points

4. An error could have been introduced while copying the results from the computer software to the web-page.

Our aim is to reduce these risks of errors. For this purpose we generate automatically the description of the points and we provide formal proofs. We created a certified version of ETC, such that each property of a point come along with a computer checked proof of its validity with the necessary non degeneracy assumptions.

The work closest to ours is the work of Deko Dekov who uses a computer program to generate theorems in triangle geometry<sup>5</sup> and present the results as a *Computer Generated Encyclopedia of Euclidean Geometry*<sup>6</sup>. This work differs from ours because the proofs of the results are not given and the facts are not checked using a proof assistant.

In the first section we provide an overview of the encyclopedia from the point of view of formal theorem proving. In the second section we describe our formalization using the Coq proof assistant of the required concepts. In the third section we explain how we obtained automatically thousands of properties and how we proved them.

## 1. The Encyclopedia of Triangle Centers

The triangle centers defined in the encyclopedia of Clark Kimberling are various. The first centers are well known points studied for centuries. The other points are defined by various means: most are defined by a traditional geometric constructions, some are defined as minimizer of some geometric quantity, there is even a point defined as a solution of an equation related to a physical property ( $X_{5626}$  is the center of maximal electrostatic potential inside a triangle  $ABC$  having a homogeneous surface charge distribution). Table 1 provides the list of the first 5 points in ETC.  $a$ ,  $b$  and  $c$  denote the side length of triangle  $ABC$  ( $a = |BC|$ ,  $b = |AC|$  and  $c = |AB|$ ). Figure 1 depicts the first 4000 points of ETC.

The points are defined by different means but are always manipulated using their homogeneous coordinates relative to the triangle  $ABC$ . The coordinates used are either the barycentric coordinates or the trilinear coordinates which are the relative distance from the point to the sides of the triangle  $ABC$ . These coordinates are expressed using the length of the sides:  $a$ ,  $b$ ,  $c$  or equivalently the measure of the three angles of triangle  $ABC$ . A given point  $P$  has trilinear coordinates any triple of the form  $ka' : kb' : kc'$  with  $k \neq 0$  and  $a'$ ,  $b'$  and  $c'$  denote the distances from  $P$  to the side lines of  $ABC$  (Fig.2).

Kimberling defined a triangle center as a point  $X$  such that there is a function  $f$  of the side lengths of triangle  $ABC$  such that:  $X = f(a, b, c) : f(b, c, a) : f(c, a, b)$  and  $f$  should be homogeneous, there must be an integer  $n$  such that:  $\forall a, b, c, t, t > 0 \Rightarrow f(ta, tb, tc) = t^n f(a, b, c)$  and symmetric in the sense that  $f(a, b, c) = f(a, c, b)$ . The fact that  $f$  is a function of the side length ensure that the triangle centers are preserved by isometries (as the side lengths are preserved by isometries). The fact that  $f$  is homogeneous moreover ensures that triangle centers of similar triangles are preserved.

<sup>5</sup><http://www.ddekov.eu/j/contents.htm>

<sup>6</sup><http://www.ddekov.eu/e1/>




FIGURE 1. The first 4000 centers displayed (sketch generated using GeoGebra 5.0).




FIGURE 2. Trilinear Coordinates

To give to the reader an idea of what can be found in Clark Kimberling's encyclopedia, we reproduce here the entry of the orthocenter (Fig. 3). For most points in the encyclopedia the trilinear and barycentric coordinates are given (often with several equivalent formulas). Then, some comments are given about the point in natural language. Finally, we have a list of properties. The lines that the point belongs to are listed. Then a list of relations between the points is given. For instance,  $X_4$  is the midpoint of the segment  $X_3X_{382}$ . The amount of information available differs from one point to the other.

### Categories of points

We can distinguish triangle centers depending on how they can be expressed using trilinear coordinates, this will be important to obtain proofs.

About 87% of the points can be defined by trilinear coordinates which are rational fractions in  $a$ ,  $b$  and  $c$ . The definition of quite a few points (about 12%) involve square roots, either square roots of constants or also square roots of rational fractions in  $a$ ,  $b$  and  $c$ . There are only a few occurrences of nested square roots in the original list of coordinates collected by Peter Moses. One definition involve a constant ( $\sqrt{25 - 10\sqrt{5}}$  in the definition of  $X_{1139}$ ) but the other occurrences can be simplified to

**X(4) = ORTHOCENTER**

Trilinears  $\sec A : \sec B : \sec C$   
 $= \cos A - \sin B \sin C : \cos B - \sin C \sin A : \cos C - \sin A \sin B$   
 $= \cos A - \cos(B - C) : \cos B - \cos(C - A) : \cos C - \cos(A - B)$   
 $= \sin B \sin C - \cos(B - C) : \sin C \sin A - \cos(C - A) : \sin A \sin B - \cos(A - B)$

Barycentrics  $\tan A : \tan B : \tan C$

As a point on the Euler line, X(4) has Shinagawa coefficients (0, 1).

X(4) is the point of concurrence of the altitudes of ABC.

If you have The Geometer's Sketchpad, you can view [Orthocenter](#).

X(4) and the vertices A,B,C comprise an orthocentric system, defined as a set of four points, one of which is the orthocenter of the triangle of the other three (so that each is the orthocenter of the other three

Suppose P is not on a sideline of ABC. Let A'B'C' be the cevian triangle of P. Let D,E,F be the circles having diameters AA', BB',CC'. The radical center of D,E,F is X(4). (Floor van Lamoen, Hyacinthos #

**Ross Honsberger, *Episodes in Nineteenth and Twentieth Century Euclidean Geometry*, Mathematical Association of America, 1995. Chapter 2: The Orthocenter.**

X(4) lies on the Thomson cubic, the Darboux cubic, the Napoleon cubic, the Neuberg cubic, and these lines:  
 1,33 2,3 6,53 7,273 8,72 9,10 11,56 12,55 13,61 14,62 15,17 16,18 32,98 35,498 36,499 37,1841 39,232 42,1860 46,90 48,1881 49,156 51,185 52,68 54,184 57,84 64,1853 +  
 120,1292 121,1293 122,1294 123,1295 126,1296 127,1289 128,930 129,1303 130,1298 131,135 137,933 141,1350 145,149 147,148 150,152 155,254 162,270 165,1698 171,601 19,  
 487,489 488,490 496,999 512,879 523,1552 542,576 569,1179 572,1474 574,1506 575,598 579,1713 580,1714 590,1151 608,1518 615,1152 616,627 617,628 653,1156 774,1254 +  
 1161,1163 1251,1832 1329,1376 1340,1348 1341,1349 1385,1538 1430,1468 1499,1550 1715,1730 1716,1721 1717,1718 1726,1782 3065,3464 3347,3472 3348,3355

X(4) is the {X(3),X(5)}-harmonic conjugate of X(2). For a list of other harmonic conjugates of X(4), click **Tables** at the top of this page.

X(4) = midpoint of X(i) and X(j) for these (i,j): (3,382), (146,3448), (147,148), (149,153), (150,152)

X(4) = reflection of X(i) in X(j) for these (i,j): (1,946), (2,381), (3,5), (5,546), (8,355), (20,3), (24,235), (25,1596), (40,10), (69,1352), (74,125), (98,115), (99,114), (100,119), (101,118), (102,124), (103,111), (114,1313), (1141,137), (1292,120), (1293,121), (1294,122), (1295,123), (1296,126), (1297,127), (1298,130), (1299,135), (1300,136), (1303, 129), (1350,141), (1593,1595)

X(4) = isogonal conjugate of X(3)  
 X(4) = isotomic conjugate of X(69)  
 X(4) = cyclocevian conjugate of X(2)  
 X(4) = inverse-in-circumcircle of X(186)  
 X(4) = inverse-in-nine-point-circle of X(403)  
 X(4) = complement of X(20)  
 X(4) = anticomplement of X(3)  
 X(4) = complementary conjugate of X(2883)  
 X(4) = anticomplementary conjugate of X(20)  
 X(4) = eigencenter of cevian triangle of X(i) for I = 1, 88, 162  
 X(4) = eigencenter of anticevian triangle of X(i) for I = 1, 44, 513

X(4) = X(i)-Ceva conjugate of X(j) for these (i,j):  
 (7,196), (27,19), (29,1), (92,281), (107,523), (264,2), (273,278), (275,6), (286,92), (393, 459)

X(4) = cevapoint of X(i) and X(j) for these (i,j):  
 (1,46), (2,193), (3,155), (5,52), (6,25), (11,513), (19,33), (30,113), (34,208), (37,209), (39,211), (51,53), (65,225), (114,511), (115,512), (116,514), (117,515), (118,516), (119,517), (120,518), (121,519), (1

X(4) = X(i)-cross conjugate of X(j) for these (i,j):  
 (3,254), (6,2), (19,278), (25,393), (33,281), (51,6), (52,24), (65,1), (113,403), (125,523), (185,3), (193,459), (225,158), (389,54), (397,17), (398,18), (407,225), (427,264), (512,112), (513,108), (523,107)

X(4) = crosspoint of X(i) and X(j) for these (i,j): (2,253), (7,189), (27,286), (92,273)

X(4) = crosssum of X(i) and X(j) for these (i,j):  
 (4,1075), (6,154), (25,1033), (48,212), (55,198), (56,1035), (71,228), (184,577), (185,417), (216,418)

X(4) = crossdifference of any two points on line X(520)X(647)

X(4) = X(i)-Hirst inverse of X(j) for these (i,j):  
 (1,243), (2,297), (3,350), (19,242), (21,425), (24,421), (25,419), (27,423), (28,422), (29,415), (420,427), (424,451), (459,460), (470,471), (1249,1503)

X(4) = X(i)-aleph conjugate of X(j) for these (i,j): (1,1047), (29,4)

X(4) = X(i)-beth conjugate of X(j) for these (i,j):  
 (4,34), (8,40), (29,4), (162,56), (318,318), (811,331)

FIGURE 3. An example of one of the 7000 entries in Clark Kimberling's encyclopedia: point  $X_4$  orthocenter.

suppress the nested radicals. Some points (for instance  $X_{506}$  and  $X_{507}$ ) contain cube roots or fourth roots ( $\cdot^{1/3}, \cdot^{1/4}$ ) The definition of about 1% of the points contains trigonometric functions which can not be eliminated : sin, cos, tan, sec, csc, sec, cot. For many points (such as the circumcenter or orthocenter) the definition using trigonometric functions is short but they can be eliminated to ease computations using the law of cosines ( $\cos(A) = (b^2 + c^2 - a^2)/(2bc)$ ).

The definition of some points make use of the value of the angles ( $A, B, C$ ) without using any trigonometric function. For example  $X_{360}$ , Hofstadter zero point, has trilinear coordinates:  $A/a : B/b : C/c$ . For very few points ( $X_{368}, X_{369}, X_{370}, X_{1144}, X_{3232}, X_{5373}, X_{5626}$ ) we do not have the explicit coordinates, thus we do not treat those points. In order to obtain smaller expressions for the coordinates, some notations (known as Conway's notations) are used, Tab. 2 lists those notations.

Symbol	Description	Definition
$\Delta$	Triangle area	$\frac{\sqrt{(a+b-c)(a-b+c)(-a+b+c)(a+b+c)}}{4}$
SW		$\frac{a^2+b^2+c^2}{2}$
sa		$\frac{b+c-a}{2}$
sb		$\frac{c+a-b}{2}$
sc		$\frac{a+b-c}{2}$
s		Half-perimeter
J		$\frac{\sqrt{a^6-a^4b^2-a^2b^4+b^6-a^4c^2+3a^2b^2c^2-b^4c^2-a^2c^4-b^2c^4+c^6}}{abc}$
e		$\sqrt{\frac{a^4-a^2b^2+b^4-a^2c^2-b^2c^2+c^4}{a^2b^2+a^2c^2+b^2c^2}}$
R	Radius of the circumcircle	$\frac{abc}{\sqrt{(a+b+c)(b+c-a)(c+a-b)(a+b-c)}}$
r	Radius of the incircle	$\frac{1}{2} \sqrt{\frac{(b+c-a)(c+a-b)(a+b-c)}{a+b+c}}$

TABLE 2. Conway's notations.

## 2. Formalization of the main concepts using Coq

In this section, we describe the formalization in Coq of the concepts we need. The results given in this section are stated and proved interactively using the Coq proof assistant (even if we use some automatic tactics). In the next sections, we will describe the facts which have been found and proved completely automatically.

For the Coq formalization we start with the type of the points, we use a record with three real numbers to represent the homogeneous coordinates. We don't specify in the type the exact kind of coordinate we use, it can be either trilinear or barycentric coordinates. This type will also be used to represent lines because of the duality between points and lines using homogeneous coordinates.

---

```
Record Point : Set :=
  cTriple
  {
    X : R;
    Y : R;
    Z : R
  }.

```

---

Note that a unique point can be represented using different homogeneous coordinates, hence we define point equality as:

---

```
Definition eq_points P Q := X(Q)*Y(P) = X(P)*Y(Q) /\ X(Q)*Z(P) = X(P)*Z(Q).

```

---

We also define a type class representing the given base triangle defined by three real numbers: the lengths of the sides  $a$ ,  $b$  and  $c$ . We assume that the sides are strictly positive and verify the triangle inequality.

---

```
Class triangle_theory := {
  a:R;
  b:R;
  c:R;
  apos: a > 0;
  bpos: b > 0;

```

```

cpos: c > 0;
ineq_1 : a + b - c > 0;
ineq_2 : a - b + c > 0;
ineq_3 : -a + b + c > 0
}.

```

---

In this setting, the three reference points can be defined as<sup>7</sup>:

```

Definition pointA := cTriple 1 0 0.
Definition pointB := cTriple 0 1 0.
Definition pointC := cTriple 0 0 1.

```

---

As many theorems are incorrect for points on the side lines of triangle  $ABC$ , to simplify our statements we introduce a predicate:

```

Definition is_not_on_sidelines P :=
  match P with
  | cTriple X Y Z => X <> 0 /\ Y <> 0 /\ Z <> 0
  | _ => false
end.

```

---

To ease the definition of triangle centers we define a constructor which take as an argument an homogeneous function  $h$  and produce the associated triangle center:

```

Context '{M:triangle_theory}. (* We assume we have a triangle theory. *)

Definition cPointh h := cTriple (h a b c) (h b c a) (h c a b).

```

---

Our development uses trilinear coordinates, but the definition of the coordinates of the points that Peter Moses kindly gave us, uses barycentric coordinates. Fortunately it is easy to convert from one to the other. For more convenience we define an alternative constructor to obtain trilinear coordinates from a function for barycentric coordinates:

```

Definition cPointhb h := cTriple ((h a b c)*b*c) ((h b c a)*a*c) ((h c a b)*a*b).

```

---

Using these constructors we define 6000 points, we list here the first five points:

```

Definition X_1 := let h_x_1 a b c := a in cPointhb h_x_1.
Definition X_2 := let h_x_2 a b c := 1 in cPointhb h_x_2.
Definition X_3 := let h_x_3 a b c := a^2*(a^2-b^2-c^2) in cPointhb h_x_3.
Definition X_4 := let h_x_4 a b c := (a^2+b^2-c^2)*(a^2-b^2+c^2) in cPointhb h_x_4.
Definition X_5 := let h_x_5 a b c := a^2*b^2-b^4+a^2*c^2+2*b^2*c^2-c^4 in cPointhb h_x_5

```

---

## 2.1. Incidence geometry

A very important predicate in geometry is the collinearity predicate. Using homogeneous coordinates three points are collinear if their determinant is zero, we can define the `col` predicate as:

```

Definition col A B C :=
  X(A)*Y(B)*Z(C) + Y(A)*Z(B)*X(C) + Z(A)*X(B)*Y(C)
- Z(A)*Y(B)*X(C) - Y(A)*X(B)*Z(C) - X(A)*Z(B)*Y(C) = 0.

```

---

The line going through  $P$  and  $Q$  can be defined as a triple (the three coefficients of the equation of the line):

---

<sup>7</sup>We name the points like that to keep the letters  $A B C$  for other purposes.

---

```

Definition line P Q :=
  match P,Q with
  cTriple x1 y1 z1, cTriple x2 y2 z2 =>
  cTriple (y1*z2-y2*z1) (z1*x2-z2*x1) (x1*y2-x2*y1)
  end.

```

---

Using the duality, the point of intersection of two lines can use the same definition, but we provide an alternative definition to increase readability:

---

```

Definition inter L M := line L M.

```

---

## 2.2. Euclidean Geometry

Two lines of coordinates  $(x_1, y_1, z_1)$  and  $(x_2, y_2, z_2)$  are parallel if

$$\begin{vmatrix} x_1 & y_1 & z_1 \\ x_2 & y_2 & z_2 \\ a & b & c \end{vmatrix} = 0:$$


---

```

Definition parallel L M :=
  match L,M with
  cTriple x1 y1 z1, cTriple x2 y2 z2 =>
  x1*y2*c+x2*b*z1+a*y1*z2 -a*y2*z1-x1*b*z2-x2*y1*c = 0
  end.

```

---

We can then easily prove the triangle midpoint theorem to check our definitions:

---

```

Lemma triangle_midpoint : forall A B C,
  parallel (line (midpoint A B) (midpoint A C)) (line B C).

```

---

Now, we can define a predicate for perpendicular lines. We adapt the formula which can be found in the literature [Gal10] to get rid of trigonometric functions and divisions:

---

```

Definition perp L M :=
  match L,M with
  cTriple x1 y1 z1, cTriple x2 y2 z2 =>
  2*a*b*c*(x1*x2 + y1*y2 + z1*z2)
  -(y1*z2 + y2*z1)* a* (b^2+c^2-a^2)
  -(z1*x2 + z2*x1)* b* (c^2+a^2-b^2)
  -(x1*y2 + x2*y1)* c* (a^2+b^2-c^2) = 0
  end.

```

---

Then, we can check that the circumcenter ( $X_3$  in Kimberling's numbering) belongs to the perpendicular bisectors:

---

```

Lemma circumcenter_is_on_perp_bisect :
  perp (line X_3 (midpoint pointA pointB)) (line pointA pointB).

```

---

We define the foot of the altitudes:

---

```

Definition T_A := inter (line pointB pointC) (line pointA X_4).
Definition T_B := inter (line pointA pointC) (line pointB X_4).
Definition T_C := inter (line pointA pointB) (line pointC X_4).

```

---

With these definitions we can check that the orthocenter is properly defined:

---

```

Lemma orthocenter_check :
  perp (line X_4 T_A) (line pointB pointC) /\

```






FIGURE 4. Cevian triangle and pedal triangle.

---

```

perp (line X_4 T_B) (line pointA pointC) /\
perp (line X_4 T_C) (line pointA pointB).

```

---

### 2.3. About triangles

**Cevian and anti-Cevian triangles.** The concept of Cevian triangles will be useful in the rest of this paper. In this paragraph we define the Cevian and anti-Cevian triangles of a point  $P$ . We adopt a definition based on coordinates and we will prove afterward that the definition is correct. Recall the Cevian triangle of a point  $P$  in a triangle  $ABC$  is the triangle formed by the intersections of the three lines joining the vertexes of  $ABC$  to  $P$  and the sides of  $ABC$  (Fig. 4).  $A'B'C'$  is the anti-Cevian triangle of  $P$  in  $ABC$ , if  $ABC$  is the Cevian-triangle of  $P$  in  $ABC$ .

---

```

Definition cevian_triangle P :=
  match P with cTriple p q r =>
    (cTriple 0 q r, cTriple p 0 r, cTriple p q 0)
  end.

```

```

Definition anti_cevian_triangle P :=
  match P with cTriple p q r =>
    (cTriple (-p) q r, cTriple p (-q) r, cTriple p q (-r))
  end.

```

---

But these definitions allow us to define only the Cevian triangle of  $P$  w.r.t. the base triangle. If we want to consider the Cevian-triangle of  $P$  in any triangle  $ABC$ , we need a more general definition:

---

```

Definition cevian_triangle_gen P A B C :=
  ((inter (line A P) (line B C)),
   (inter (line B P) (line A C)),
   (inter (line C P) (line A B))).

```

---

We can convince ourselves that particular case for the base triangle  $ABC$  is consistent with the general definition:

---

```

Lemma cevian_triangle_ok : forall P,
  eq_triangles (cevian_triangle P) (cevian_triangle_gen P pointA pointB pointC).

```

---

We can prove also that we really defined the Cevian triangle:

---

```

Lemma cevian_triangle_gen_ok :
forall P A B C,
  match cevian_triangle_gen P A B C with
    (X,Y,Z) => col A P X /\ col B P Y /\ col C P Z /\
              col X B C /\ col Y A C /\ col Z A B
  end.

```

---

Finally, we can name some well known triangles:

---

```

Definition incentral_triangle := cevian_triangle X_1.
Definition medial_triangle    := cevian_triangle X_2.
Definition orthic_triangle    := cevian_triangle X_4.
Definition intouch_triangle   := cevian_triangle X_7.
Definition extouch_triangle   := cevian_triangle X_8.

```

---

**Pedal triangle.** The Pedal triangle  $A_P B_P C_P$  of a point  $P$  is the triangle formed by the projections of  $P$  on the side lines (Fig.4), we can define its coordinates:

---

```

Definition pedal_triangle P :=
match P with cTriple p q r =>
(cTriple 0 (q+p*(-a^2-b^2+c^2)/(-2*a*b)) (r + p*(-a^2+b^2-c^2)/(-2*a*c)),
 cTriple (p + q*(-a^2-b^2+c^2)/(-2*a*b)) 0 (r+q*(a^2-b^2-c^2)/(-2*b*c)),
 cTriple (p + r*(-a^2+b^2-c^2)/(-2*a*c)) (q+r*(a^2-b^2-c^2)/(-2*b*c)) 0)
end.

```

---

As previously, we prove afterward that our definition make sense:

---

```

Lemma pedal_triangle_ok : forall P,
match pedal_triangle P with
  (A1,B1,C1) => perp (line P A1) (line pointB pointC) /\
                perp (line P B1) (line pointA pointC) /\
                perp (line P C1) (line pointA pointB)
end.

```

---

Obviously, the Cevian triangle of the orthocenter is the same as the pedal triangle:

---

```

Lemma cevian_triangle_of_orthocenter_is_pedal :
eq_triangles (cevian_triangle X_4) (pedal_triangle X_4).

```

---

Moreover, the medial triangle is the pedal triangle of the circumcenter:

---

```

Lemma pedal_triangle_of_circumcenter_is_medial :
eq_triangles medial_triangle (pedal_triangle X_3).

```

---

The intouch triangle is the pedal triangle of the incenter:

---

```

Lemma pedal_triangle_of_incenter_is_intouch :
eq_triangles intouch_triangle (pedal_triangle X_1).

```

---

**Perspective triangles.** Two triangles  $ABC$  and  $A'B'C'$  are said to be perspective from  $O$  if their three pairs of corresponding vertexes are joined by lines which meet in  $O$  (Fig. 5):

---

```

Definition is_perspector O T1 T2 :=
match T1,T2 with
  (A,B,C), (A',B',C') =>

```

---




FIGURE 5. Perspective triangles.

---

```

col 0 A A' /\ col 0 B B' /\ col 0 C C'
end.

```

---

For example,  $P$  is the perspector of any triangle  $ABC$  and its Cevian triangle in  $ABC$ :

---

```

Lemma cevian_triangle_perspector :
forall A B C P, is_perspector P (cevian_triangle_gen P A B C) (A,B,C).

```

---

#### 2.4. The geometric transformations

We define the geometric transformations using homogeneous coordinates. Table 3 lists all the transformations we can deal with. The definitions of these transformations are taken from Kimberling's glossary or Paul Yiu's book [Yiu02]. We could have given the definitions using geometric constructions, but we adopt algebraic definitions for two reasons. First, with algebraic definition, the proofs are relatively easy to obtain. Second, the geometric definitions often require a proof to certify the definition make sense. For example, the geometric definition of the centroid requires to prove that the medians intersect. Still, we want to make sure we did not make any mistake while giving our definitions. To check our definitions, we prove the geometric properties.

**Ceva conjugate.** As a first example, let's consider the  $P$ -Ceva conjugate of  $Q$ . By definition it is the perspector of the Cevian triangle of  $P$  and the anti-Cevian triangle of  $Q$  but we define it as:

---

```

Definition ceva_conjugate P U :=
match P,U with
cTriple p q r, cTriple u v w =>
cTriple (u*(-q*r*u+r*p*v+p*q*w)) (v*(q*r*u-r*p*v+p*q*w)) (w*(q*r*u+r*p*v-p*q*w))
end.

```

---

To prove that our definition is consistent with the geometric definition, we can prove that the  $P$ -Ceva conjugate of  $Q$  is the perspector of the Cevian triangle of  $P$  and anti-Cevian triangle of  $Q$  (Fig. 6):

---

```

Lemma ceva_conjugate_property :
forall P Q, is_perspector (ceva_conjugate P Q) (cevian_triangle P) (anti_cevian_triangle Q).

```

---

Name#Arity	Definition		
anticomplement#1	$\frac{-ap+bq+cr}{a}$	$\frac{ap-bq+cr}{b}$	$\frac{ap+bq-cr}{c}$
ceva_conjugate#2	$u(-qru + rpv + pqw)$	$v(qru - rpv + pqw)$	$w(qru + rpv - pqw)$
complement#1	$\frac{bq+cr}{a}$	$\frac{ap+cr}{b}$	$\frac{ap+bq}{c}$
cross_conjugate#2	$\frac{u}{-pvw+qwu+ruv}$	$\frac{v}{pvw-quu+ruv}$	$\frac{w}{pvw+qwu-ruv}$
cross_difference#2	$qw - rv$	$ru - pw$	$pv - qu$
cross_sum#2	$qw + rv$	$ru + pw$	$pv + qu$
cross_point#2	$pu(rv + qw)$	$qv(pw + ru)$	$rw(qu + pv)$
daleth_conjugate#2	let $h(p, q, r, u, v, w) = p(\frac{v}{q} - \frac{w}{r})^2 + u(\frac{2u}{p} - \frac{v}{q} - \frac{w}{r})$ in		
	$h(p, q, r, u, v, w)$	$h(q, r, p, v, w, u)$	$h(r, p, q, w, u, v)$
hirst_inverse#2	$qu^2 - vwp^2$	$rpv^2 - wuq^2$	$pqw^2 - uvr^2$
iso_conjugate#2	$qrvw$	$rpwu$	$pquv$
isogonal_conjugate#1	$\frac{1}{p}$	$\frac{1}{q}$	$\frac{1}{r}$
isotomic_conjugate#1	$\frac{1}{pa^2}$	$\frac{1}{qb^2}$	$\frac{1}{rc^2}$
line_conjugate#2	$p(v^2 + w^2) -$ $u(qv + rw)$	$q(w^2 + u^2) -$ $v(rw + pu)$	$r(u^2 + v^2) -$ $w(pu + qv)$
midpoint#2	let $h = ap + bq + cr$ in let $k = au + bv + cw$ in		
	$kp + hu$	$kq + hv$	$kr + hw$
reflection#2	$(ap - bq - cr)u +$ $2p(bv + cw)$	$(bq - ap - cr)v +$ $2q(au + cw)$	$(cr - ap - bq)w +$ $2r(au + bv)$
waw_conjugate#2	let $h(p, q, r, x, y, z) = p(x^2q^2r^2 + 2p^2(ry - qz)^2 - pqr^2xy - pq^2rxz)$ in		
	$h(p, q, r, u, v, w)$	$h(q, r, p, v, w, u)$	$h(r, p, q, w, u, v)$
zayin_conjugate #2	let $h(p, q, r, x, y, z) = p(y + z)^2 - ry^2 - qz^2 + (p - r)xy + (p - q)xz$ in		
	$h(p, q, r, u, v, w)$	$h(q, r, p, v, w, u)$	$h(r, p, q, w, u, v)$

TABLE 3. The transformations formalized.

**Isotomic conjugate.** A second example is the isotomic conjugate. Given a point  $P$ , let  $A_1B_1C_1$  be the Cevian triangle of  $P$  in  $ABC$ . Let  $A_2$  (resp.  $B_2, C_2$ ) be reflection of  $A_1$  (resp.  $B_1, C_1$ ) in the midpoint of segment  $BC$  (resp.  $AC, AB$ ). The lines  $AA_2, BB_2$  and  $CC_2$  meet at a point  $P'$  called the isotomic conjugate of  $P$  (Fig. 7). We define it using coordinates as:

---

```

Definition isotomic_conjugate P :=
  match P with
    cTriple p q r =>
      cTriple (1/(p*a^2)) (1/(q*b^2)) (1/(r*c^2))
  end.

```

---




FIGURE 6. The Ceva conjugate.




FIGURE 7. The isotomic conjugate.

Then to prove that our definition is correct, we express the fact that if we reflect the Cevian triangle of  $P$  in the midpoints of the sides of  $ABC$  we obtain the Cevian triangle of its isotomic conjugate. Note that we need to assume that  $P$  is not on any sideline of triangle  $ABC$ .

---

```

Lemma isotomic_conjugate_property :
forall P,
  is_not_on_sidelines P ->
  match cevian_triangle P with
  (A1,B1,C1) =>
  match cevian_triangle (isotomic_conjugate P) with
  (A2,B2,C2) => eq_points (midpoint pointA pointB) (midpoint C1 C2)
  end
  
```

---

Easily we can show also that this transformation is an involution:

---

```

Lemma isotomic_involution : forall P,
  is_not_on_sidelines P ->
  eq_points (isotomic_conjugate (isotomic_conjugate P)) P.

```

---

**Complement and anti-complement.**  $U$  is the complement of a point  $P$  if it belongs to line  $X_2P$  (where  $X_2$  is the centroid) and  $X_2$  trisects segment  $PU$  and  $X_2$  is closer to  $P$  than  $U$ . We define it as:

---

```

Definition complement P :=
  match P with
  | cTriple p q r =>
  | cTriple ((b*q+c*r)/a) ((a*p+c*r)/b) ((a*p+b*q)/c)
  end.

```

```

Definition anticomplement P :=
  match P with
  | cTriple p q r =>
  | cTriple ((-a*p+b*q+c*r)/a) ((a*p-b*q+c*r)/b) ((a*p+b*q-c*r)/c)
  end.

```

---

We can prove that they are inverse function of each-other:

---

```

Lemma anticomplement_complement_property :
  forall P, eq_points (anticomplement (complement P)) P.

```

---

and that the centroid belongs to the line:

---

```

Lemma complement_property :
  forall P, col P (complement P) X_2.

```

---

**Reflection and midpoint.** Following the formulas given in Tab. 3, we can define the midpoint of a segment and the reflection of point wrt. another point, then we can check that:

---

```

Lemma reflection_midpoint : forall A B,
  eq_points (midpoint A (reflection A B)) B.

```

---

As an example, we can also prove that the center of the nine point circle is the midpoint of the segment  $X_3X_4$  ( $X_3$  is the circumcenter and  $X_4$  the orthocenter):

---

```

Lemma nine_point_circle_bisects_OH:
  eq_points X_5 (midpoint X_3 X_4).

```

---

### 3. Finding conjectures and automating their proofs

In the previous section, we listed our definitions and presented some results that we proved interactively just to verify our definitions. In this section we describe the algorithms we used to find and prove automatically some properties about triangle centers.

#### 3.1. Finding properties

There are so many properties presented in Clark Kimberling's encyclopedia that obtaining an exhaustive list of these properties is difficult. Indeed, it would require to parse the information presented on the web pages, but the computer generated pieces of information are inter-mixed with human generated comments in such a way that it would make this parsing phase non-trivial. Hence, we chose not to check the properties displayed on the page, but to generate ourself these properties. Fortunately

most of the properties displayed in the encyclopedia are relatively easy to obtain by computer means. In this section, we describe how we obtained these properties.

We address two main kind of problems:

1. Find all the collinearity properties in the list of points. Namely find all the triples  $(i, j, k)$  such that  $Col(X_i, X_j, X_k)$ .
2. Given a function  $f$  which takes  $n$  points as arguments and return a point, find all  $n + 1$ -tuples  $(i, j_1, \dots, j_n)$  such that  $X_i = f(X_{j_1}, \dots, X_{j_n})$ .

We want to find properties which are valid in any triangle, but to simplify the computations we consider one example triangle. If the property does not hold in this triangle, it does not hold in general, if it holds in this particular triangle then there are good chance that it holds in general. In practice the coincidences are very rare (but Francisco Javier recently pointed that points  $X_{3875}$  and  $X_{7292}$  have exactly the same coordinate in the triangle  $(a, b, c) = (6, 9, 13)$ ).

**3.1.1. Finding Collinearity Properties.** Let  $n$  be the number of points in ETC. We want to find all the triples  $(i, j, k)$  such that  $Col(X_i, X_j, X_k)$ . A naïve algorithm would require  $\mathcal{O}(n^3)$  operations. We implemented a better algorithm by using an associative array  $T$  to store for each line the set of points belonging to this line. For all every couple of points  $(X_i, X_j)$  we compute the equation  $e$  of the line  $X_i X_j$  and we add the points  $X_i$  and  $X_j$  to  $T[e]$ . This algorithm has a complexity  $\mathcal{O}(n^2)$  on average.

The main difficulty concerns the numerical computations. Our first prototype implemented using Maple<sup>TM</sup> was producing both false positive and false negative: some points appeared to be collinear whereas there were not (even in the special case of the chosen triangle) and also some points which are collinear were reported not to be collinear. This is due to the fact that Maple<sup>TM</sup> (or other similar software) does not compute itself the precision needed to obtain a significant result. In a second prototype, we use what we call a symbolic-numeric approach. Instead of representing the coordinates of the points using big-floats approximations, we used symbolic expressions involving constants, arithmetic operations  $(+, -, *, /)$ , square roots and trigonometric functions. Deciding equality of two expressions involving radicals is possible but not trivial: for example  $\sqrt{3} + 2\sqrt{2} = 1 - \sqrt{2}$ . To decide such an equality involving only constants we use the `radnormal` function of Maple<sup>TM</sup>. But we are not aware of any algorithm dealing with the trigonometric functions. To prove the facts, we need an algorithm to normalize symbolic expressions containing radicals, an algorithm has been proposed by Rybowicz which generalize an algorithm introduced by Zippel [Zip85, Ryb03] but we are not aware of any implementation in a CAS.

**3.1.2. Finding Properties About Transformations.** To find the properties about transformations we use a brute-force algorithm. Recall that, given a function  $f$  which takes  $n$  points as arguments and return a point, we need to find all  $n + 1$ -tuples  $(i, j_1, \dots, j_n)$  such that  $X_i = f(X_{j_1}, \dots, X_{j_n})$ . For every tuple of points  $(X_{j_1}, \dots, X_{j_n})$  we compute  $X = f(X_{j_1}, \dots, X_{j_n})$  and we check if  $X$  belongs to the (sorted) list of points. The complexity of this algorithm is  $\mathcal{O}(k^n \ln(k))$  where  $k$  is the number of points in the encyclopedia.

## 3.2. Proving the properties

To prove the different properties in Coq, we make heavy use of the `ring` and `field` tactics of Coq first introduced by Delahaye and Mayero and improved by Grégoire and Mahboubi [DM01, GM05]. The idea of these reflexive tactics is the following: instead of convincing Coq that two expressions representing the same polynomial are equal by repeated applications of commutativity, associativity and distributivity properties, an algorithm to compute a normal form is written in Coq's language and proved correct. For more information about how to develop reflexive tactics see Chapter 16 of [BC04].

Proving properties involving only points whose coordinates are rational fractions is straightforward, we just need to unfold all the definitions to obtain a goal which can be solved by the field tactic. We do not treat the points whose coordinates involve trigonometric functions. For the points with square roots in their coordinates we use the following heuristics:

Name	Number of facts proved	Number of conjectures	Percentage
ceva_conjugate	7592	13091	58%
complement	906	932	97%
cross_conjugate	2249	5483	41%
cross_sum	4989	8552	58%
cross_point	3381	8052	41%
daleth_conjugate	8939	16216	55%
hirst_inverse	9130	16383	56%
isogonal_conjugate	1905	1954	97%
isotomic_conjugate	604	616	98%
line_conjugate	40136	57074	70%
waw_conjugate	83	805	10%
<i>total</i>	<i>79914</i>	<i>129158</i>	<i>62%</i>

TABLE 4. Statistics

1. We try to simplify square roots by applying the following lemma repeatedly (where  $S\ n$  stands for  $n + 1$ ):

---

```
Lemma sqrt_pow : forall x n, 0 <= x ->
  (sqrt x)^(S (S n)) = x*(sqrt x)^n.
```

---

2. We normalize all notations before unfolding them. For each notation, we have a lemma stating a permutation property, such as:

---

```
Lemma Delta_perm_1 : DeltaMaj a b c = DeltaMaj b c a.
```

---

3. We normalize all square roots of rational fractions appearing in the expression. We have a tactic `normalize_all_sqrt` which computes the normal form of every square root appearing in the goal. Here is an example of its use:

---

```
Goal sqrt ((a+b)^2) + sqrt(a+b) = sqrt(b+a)+sqrt(a^2+b^2+2*a*b).
Proof.
  normalize_all_sqrt.
  ring.
Qed.
```

---

**Non degeneracy conditions.** The non degeneracy conditions needed for proving the properties are that all the denominators appearing in the statements are different from zero and the the argument of any square root is positive, they can be generated automatically.

### 3.3. Results

Figure 8 shows a snapshot our website displaying the pieces of information about the point  $X_1$ . Each fact is displayed in three colors:

**Green.** We have a proof verified by the Coq proof assistant of this fact (assuming the non degeneracy conditions).

**Orange.** We have a Maple<sup>TM</sup> script to check this fact.

**Red.** This result is only a conjecture obtained by numerical computations.

Table 4 shows the number of conjectures and the number of properties proved. It takes about 20 computer.days for Coq to compile these files.





FIGURE 8. A snapshot of the first entry of the Certified ETC.

## Conclusion

We presented our first attempts toward a certified version of the encyclopedia of triangle centers. This experiment shows that the approach based on trilinear coordinates is efficient and can be used within a proof assistant. Compared to other automated deduction approaches such as Wu's or Gröbner's bases methods [Kap86, CG90, Wan95], this approach consists in keeping always the explicit coordinates of the points, in some sense the triangulation of the algebraic system is computed in advance. This saves a lot of computations in this context where we want to prove many properties about the same given points. This work also demonstrates that the implementation of the Coq proof assistant scales, and can manage thousands of definitions.

This work can be extended in several directions.

First, it would be interesting to connect our proof about trilinear coordinates to our development in the context of the geometry of Tarski [Nar07, BN12, BNSB14]. For example we could prove that the definition of the collinearity predicate given in this paper is equivalent with the geometric definition given in the context of Tarski's axiom system.

Second, we could study how to find and prove automatically more properties. For example we could enhance our algorithm to find properties of some well known triangles, such as *'the incenter is the circumcenter of the intouch triangle'*.

Third, we could improve the user interface. It would be interesting to have an interactive website allowing the user to ask questions in a simple syntax, to find automatically if a theorem is already known and to propose additions to the encyclopedia.

Fourth, we could generalize this work and try to certify the results presented in the encyclopedia of conics and encyclopedia of quadri-figures.

## Availability

The website is available here:

<http://dpt-info.u-strasbg.fr/~narboux/CETC/about.html>

## Thanks

We thank Peter Moses for providing a file with barycentric coordinates of the points. We thank Pascal Schreck for his comments about a draft of this paper.

## References

- [BC04] Yves Bertot and Pierre Castéran. *Interactive Theorem Proving and Program Development, Coq'Art: The Calculus of Inductive Constructions*. Texts in Theoretical Computer Science. An EATCS Series. Springer, 2004.
- [BN12] Gabriel Braun and Julien Narboux. From Tarski to Hilbert. In Tetsuo Ida and Jacques Fleuriot, editors, *Post-proceedings of Automated Deduction in Geometry 2012*, volume 7993 of *LNCS*, pages 89–109, Edinburgh, United Kingdom, September 2012. Jacques Fleuriot, Springer.
- [BNSB14] Pierre Boutry, Julien Narboux, Pascal Schreck, and Gabriel Braun. Using small scale automation to improve both accessibility and readability of formal proofs in geometry. In Francisco Botana and Pedro Quaresma, editors, *Automated Deduction in Geometry 2014*, Proceedings of ADG 2014, pages 1–19, Coimbra, Portugal, July 2014.
- [CG90] Shang-Ching Chou and Xiao-Shan Gao. Ritt-Wu's Decomposition Algorithm and Geometry Theorem Proving. In Mark E. Stickel, editor, *CADE*, volume 449 of *Lecture Notes in Computer Science*, pages 207–220. Springer, 1990.
- [DM01] David Delahaye and Micaela Mayero. Field, une procédure de décision pour les nombres réels en coq. In *JFLA*, pages 33–48, 2001.
- [Gal10] William Gallatly. *The modern geometry of the triangle*. London, F. Hodgson, 1910.
- [GM05] Benjamin Grégoire and Assia Mahboubi. Proving equalities in a commutative ring done right in coq. In Joe Hurd and Thomas F. Melham, editors, *Theorem Proving in Higher Order Logics, 18th International Conference, TPHOLs 2005, Oxford, UK, August 22-25, 2005, Proceedings*, volume 3603 of *Lecture Notes in Computer Science*, pages 98–113. Springer, 2005.
- [Kap86] Deepak Kapur. Geometry Theorem Proving using Hilbert's Nullstellensatz. In *SYMSAC '86: Proceedings of the fifth ACM symposium on Symbolic and algebraic computation*, pages 202–208, New York, NY, USA, 1986. ACM Press.
- [Kim93] Clark Kimberling. Triangle centers as functions. *Journal of Mathematics*, 23(4), 1993.
- [Kim94] Clark Kimberling. Central points and central lines in the plane of a triangle. *Mathematics Magazine*, pages 163–187, 1994.
- [Kim98] Clark Kimberling. *Triangle centers and central triangles*. 1998.
- [Nar07] Julien Narboux. Mechanical Theorem Proving in Tarski's geometry. In Francisco Botana Eugenio Roanes Lozano, editor, *Post-proceedings of Automated Deduction in Geometry 2006*, volume 4869 of *LNCS*, pages 139–156, Pontevedra, Spain, 2007. Francisco Botana, Springer.
- [Ryb03] Marc Rybowicz. On the normalization of numbers and functions defined by radicals. *Journal of Symbolic Computation*, 35(6):651–672, 2003.
- [Wan95] Dongming Wang. Elimination Procedures for Mechanical Theorem Proving in Geometry. *Ann. Math. Artif. Intell.*, 13(1-2):1–24, 1995.
- [Yiu02] Paul Yiu. *Introduction to the geometry of triangle*. 2002.
- [Zip85] Richard Zippel. Simplification of expressions involving radicals. *Journal of Symbolic Computation*, 1(2):189–210, 1985.

Julien Narboux and David Braun

e-mail: [narboux@unistra.fr](mailto:narboux@unistra.fr) [david.braun@etu.unistra.fr](mailto:david.braun@etu.unistra.fr)

ICube, UMR 7357 CNRS, University of Strasbourg

Pôle API, Bd Sébastien Brant, BP 10413, 67412 Illkirch, France